

Лабораторная работа № 2

Плоская задача о нагреве и охлаждении термоупругого тела

Цель работы.

1. Изучить основные особенности решения задач нестационарных термоупругости в ANSYS APDL на примере плоской задачи о медленном нагреве и охлаждении тела в форме буквы «R» (программа FE_Mod_CP_2.inp). Определить возможности решения задач с неполной связанностью механических и температурных полей и с пренебрежением динамических слагаемых в уравнениях движения.
2. Составить программу на языке APDL ANSYS для индивидуальной нестационарной задачи для двумерного термоупругого тела в условиях плоской деформации.
3. Провести расчеты, проанализировать результаты и оформить отчет.

Описание задачи.

Рассматривается нестационарная задача термоупругости для плоской фигуры в форме латинской буквы «R». Геометрия области, отнесенной к декартовой системе координат Oxy , показана на рис. 1. Размеры тела определяются следующими параметрами: $HL=0.14$ м (высота); $WL=0.08$ м (ширина); $WWL=0.02$ м (ширина стенок); $HAL=0.07$ м (дополнительный размер по высоте). Криволинейные части фигуры описываются частями эллипсов с параметром эллиптичности $PEL=(HAL/2)/(WL-WWL)$, который задает отношение полуоси эллипса по оси Y к полуоси эллипса по оси X .

Примем, что материалом области является сталь (изотропный материал) с модулем Юнга $E=2 \cdot 10^{11}$ Н/м², коэффициентом Пуассона $\nu=0.29$, плотностью $\rho=7800$ кг/м³, коэффициентом теплопроводности $k=K_{xx}=46.7$ Вт/(м К), коэффициентом температурного расширения $\alpha=1.51 \cdot 10^{-5}$ 1/К и удельной теплоемкостью $c_\sigma=462$ Дж/(кг К), вычисленной при постоянных напряжениях.

Предполагается, что на нижней грани температура θ все время равна 0°C . В начальный момент времени $t=0$ тело находилось в состоянии покоя. Далее с начального момента до момента времени $t=TES1=10$ на верхней грани температура линейно возрастала от нуля до 90°C . После этого на верхней грани принимаются условия конвективного теплообмена с коэффициентом теплообмена $h_f=60$ Вт/(м² К) и температурой окружающей среды $\theta_{ext}=0^\circ\text{C}$. Все остальные границы для всех рассматриваемых времен принимаются теплоизолированными,

т.е. такими, на которых нормальная компонента вектора потока тепла равна нулю: $\mathbf{n} \cdot \mathbf{q} = 0$. Нижняя граница тела все время считается жестко закрепленной.




Рис. 1. Геометрия области: (а) -- обозначения; (б) -- точки и линии

В задаче требуется определить поля температур и перемещений в интервале времени $t \in [0, TES2]$, $TES2 = 120$ с, построить графики зависимости температуры θ и перемещений u_y от времени для двух контрольных точек $\mathbf{x}_1 = [WWL, HL]$ и $\mathbf{x}_2 = [WWL, HL - HAL/2 - (WL - 2 * WWL) * PEL]$ (см. рис. 1, а), где точки помечены цифрами 1 и 2).

С целью исследования влияния связанности и динамических эффектов нужно также сравнить результаты для трех разновидностей задач:

- 1) нестационарной задачи термоупругости с полной связанностью;
- 2) нестационарной задачи термоупругости с частичной связанностью, когда задача для поля температуры не зависит от механических полей, т.е. совпадает с нестационарной с задачей теплопроводности;
- 3) задачи термоупругости с частичной связанностью, когда задача для поля температуры не зависит от механических полей, т.е. совпадает с нестационарной с задачей теплопроводности, а уравнения для перемещений рассматриваются в квазистатической постановке.

Теоретический сведения

Постановка нестационарной задачи термоупругости приведена в книге Наседкин А. В., Наседкина А.А. «Моделирование связанных задач» в разделе 1.4 на с. 45-54, определяющие соотношения для изотропных тел и задание их материальных свойств в Ansys приведены в разделе 3.2.1 на с. 132-134.

Для решения описанной выше задачи предлагается программа FE_Mod_CP_2.inp.

```
!      Файл      FE_Mod_CP_2.inp
!      Д В У М Е Р Н А Я      З А Д А Ч А
!      Т Е Р М О У П Р У Г О Е      Т Е Л О
!      Нестационарная задача

! Инвертирование фона с черного на белый
/RGB,INDEX,100,100,100,0
/RGB,INDEX,0,0,0,15

/PREP7
!   Все величины в системе Си

TES1=10    !   значение времени для окончания первого шага
TES2=120    !   значение времени для окончания второго шага
!   подшаги по времени для временных шагов
DT1=1      !   значение для шага 1
DT2=2      !   значение для шага 2

! Условия для "температурного" анализа
T_INP1=0    ! Температура на нижней границе
T_INP2=90    ! Температура на правой верхней границе (на шаге 1)
T_EXT=0     ! Температура окружающей среды для конвективных условий
H_F=60      ! Коэффициент конвективного теплообмена

! Разность температур от нуля до абсолютного нуля
TOFFST, 273
! При TOFFST=0 тепловое уравнение есть уравнение теплопроводности
! TOFFST, 0

! Материальные свойства (сталь)
E1=2.1e11 ! Модуль Юнга
NU1=0.29 ! Коэффициент Пуассона
```

```

RHO1=7.8e3 ! Плотность
K1=46.7 ! Коэффициент теплопроводности
C1=462 ! Удельная теплоемкость
ALPH1=1.51e-05 ! Коэффициент температурного расширения

MP,EX,1,E1
MP,DENS,1,RHO1
MP,NUXY,1,NU1
MP,KXX,1,K1
MP,C,1,C1
MP,ALPX,1,ALPH1

! Команды для анализа возможности не учета
! динамических слагаемых в уравнениях движения
! KAPPA=1e-6
! MP,DENS,1,RHO1*KAPPA
! MP,C,1,C1/KAPPA

ET,1,PLANE223,11,,2 ! четырехугольный КЭ с 8 узлами, степени сво-
боды UX, UY, TEMP, плоская деформация

! Геометрические размеры тела в форме буквы "R"
HL=0.14 ! Высота
WL=0.08 ! Ширина
WWL=0.02 ! Ширина стенок буквы "R"
HAL=0.07 ! Дополнительный размер по высоте

! Параметры для триангуляции
DMESH=WWL/4

PEL=(HAL/2)/(WL-WWL) ! Параметр эллиптичности для криволинейной ча-
сти
LOCAL,11,1,WWL,HL-HAL/2,,,,,PEL ! Эллиптическая система координат с
номером 11
CSYS,0 ! Переход в основную декартову систему координат

! Создание геометрии фигуры
! Внешняя часть - область 1
K,1,0,0
K,2,WWL,0
K,3,WWL,HL-HAL-WWL
K,4,WL-WWL,0
K,5,WL,0
K,6,WWL,HL-HAL

```

```

K,7,WL,HL-HAL/2
K,8,WWL,HL
K,9,0,HL
L,1,2 $ L,2,3 $ L,3,4 $ L,4,5 $ L,5,6

CSYS,11
L,6,7 $ L,7,8
CSYS,0
L,8,9 $ L,9,1
AL,1,2,3,4,5,6,7,8,9

! Отверстие - область 2
K,10,WWL,HL-HAL/2-(WL-2*WWL)*PEL
K,11,WL-WWL,HL-HAL/2
K,12,WWL,HL-HAL/2+(WL-2*WWL)*PEL
CSYS,11
L,10,11 $ L,11,12
CSYS,0
L,12,10
AL,10,11,12
ASBA,1,2 ! Вырезать из области 1 область 2

! Установка размера конечного элемента
ESIZE,DMESH

AMESH,ALL

! Условия жесткого закрепления
NSEL,S,LOC,Y,0 ! Выбор узлов внизу (по оси Y)
D,ALL,UX,0
D,ALL,UY,0
NSEL,ALL ! Вернуться к выбору всех узлов модели

! Выбор узлов для мониторинга результатов
! Узел N1_MON - в верхней части фигуры
! Y=HL, X=WWL
NSEL,S,LOC,Y,HL,HL
NSEL,R,LOC,X,WWL
*GET,N1_MON,NODE,,NUM,MIN
! Узел N2_MON - ниже по X=WWL
NSEL,S,LOC,Y,HL-HAL/2-(WL-2*WWL)*PEL
NSEL,R,LOC,X,WWL
*GET,N2_MON,NODE,,NUM,MIN
NSEL,ALL

```

FINISH

! Решение связанной задачи термоупругости

/SOLU

ANTYPE,TRANS ! Решение нестационарной задачи

KBC,0 ! Линейный рост внешних воздействий

! Коэффициент затухания в схеме интегрирования по времени

!TINTP,0.005 ! TINTP=0.005 - по умолчанию

TIME,TES1 ! Время окончания шага 1

DELTIM,DT1 ! Подшаг по времени для шага 1

NSEL,S,LOC,Y,0 ! Выбор узлов внизу (по оси Y)

D,ALL,TEMP,T_INP1 ! Задать для всех выбранных узлов TEMP=T_INP1

NSEL,S,LOC,Y,HL,HL ! Выбор узлов сверху (по оси Y)

NSEL,R,LOC,X,0,WWL

D,ALL,TEMP,T_INP2 ! Задать для всех выбранных узлов TEMP=T_INP2

NSEL,ALL

OUTRES,BASIC,ALL

SOLVE

! Шаг 2

TIME,TES2

DELTIM,DT2

KBC,1 ! Значения внешних воздействий постоянны

NSEL,S,LOC,Y,HL,HL ! Выбор узлов сверху (по оси Y)

NSEL,R,LOC,X,0,WWL

DDEL,ALL,TEMP ! Удаление условий TEMP=T_INP2

SF,ALL,CONV,H_F,T_EXT ! Конвективный теплообмен в выбранных узлах

NSEL,ALL

SOLVE

SAVE

FINISH

/POST26

/SHOW,WIN32C

/TRIAD,OFF ! Не показывать начала координат и осей

/PLOPTS,INFO,2 ! Использовать формат вывода Auto-legend для подписей

/PLOPTS,LEG2,OFF

```

/PLOPTS,LOGO,OFF      ! Логотип ANSYS не показывать в графическом виде
/PLOPTS,FRAME,OFF     ! Не показывать рамку
/PLOPTS,DATE,OFF      ! Не показывать дату
/COLOR,CURVE,WHIT,1,6 ! Цвет графиков - белый (черный при инвертации)
/COLOR,GRID,WHIT      ! Цвет сетки - белый (черный при инвертации)
/GMARKER,1,2,3        ! Для кривой 1 метка 2 - квадратики, на каждом 3-
ем значении
/GMARKER,2,4,3        ! Для кривой 2 метка 4 - крестики, на каждом 3-ем
значении
/GRID,1               ! Сетка по X и Y (полная сетка)

/XRANG,0,TES2
/AXLAB,Y,Temperature (degree) ! Подпись к оси Y
/AXLAB,X,Time (sec)          ! Подпись к оси X
NSOL,2,N1_MON,TEMP,,TEMP1    ! Переменная 2 - температура в узле
N1_MON
NSOL,3,N2_MON,TEMP,,TEMP2

NSOL,4,N1_MON,U,Y,UY1       ! Переменная 4 - перемещение UY в узле N1_MON
NSOL,5,N2_MON,U,Y,UY2

PLVAR,2,3 ! Построение графиков переменных 2, 3 от времени (пере-
менной 1)

! Снять комментарии для вывода UY в контрольных узлах
! /AXLAB,Y,UY (m)
! PLVAR,4,5

```

В данной программе решается связанная линейная задача термоупругости в рамках предположения о плоской деформации, когда $\mathbf{u} = \{u_x(x, y, t), u_y(x, y, t), 0\}$, $\theta = \theta(x, y, t)$.

Как и в примере из первой лабораторной работы, здесь вначале строится твердотельная модель, а затем, из нее – конечно-элементная. Твердотельная модель получается методом построения геометрических объектов «снизу – вверх». Для этого сначала задаются опорные точки, потом – линии (см. рис. 1, б), а затем двумерные области.

Для построения плоской модели строятся две односвязные области: первая область, ограниченная внешними линиями буквы «R», и вторая область, ограниченная внутренними линиями. Затем, с использованием булевой операции вычитания геометрических областей (команда **ASBA**) получается итоговая область

буквы «R» с отверстием. При этом криволинейные части буквы «R» строятся как части эллипсов в соответствующей эллиптической системе координат.

Конечно-элементная модель получается с использованием свободного разбиения на четырехугольные элементы PLANE223 с соответствующими опциями. Для определения густоты конечно-элементной сетки используется единственный параметр *DMESH*, задающий максимальный размер элементов. Структура конечно-элементной сетки при *DMESH=WWL/4* показана на рис. 2 (а) – элементы, б) – узлы).

На рис. 2 показаны также внешние воздействия на различных временных шагах. (Для показа граничных условий в интерактивном режиме следует из верхнего меню выполнить: *Plot->Elements* или *Plot->Nodes*, *PlotCtrls->Symbols->Отметить All Applied BCs*. Для отображения стрелками условий конвективного теплообмена в пункте *Surface Load Symbols* можно задать вывод *Convect FilmCoef* и для *Show pres and convect as* выбрать *Arrows*.)




Рис. 2. Конечно-элементная модель: (а) – конечно-элементная сетка и граничные условия на первом шаге; (б) – узлы и граничные условия на втором шаге


Решение нестационарной задачи здесь разбивается на два шага по времени $t \in [0, TES1]$ и $t \in [TES1, TES2]$. На каждом временном шаге конечно-элементная система обыкновенных дифференциальных уравнений нестационарной термоупругости интегрируется по времени с подшагом Δt_j ($\Delta t_1 = DT1 = 1$ с, $\Delta t_2 = DT2 = 2$ с) по модифицированной схеме Ньюмарка.

Отметим, что на первом и втором шагах по времени задаются различные граничные условия на верхней границе. На первом временном шаге задана темпе-

ратура (рис. 2, а) с линейным изменением по времени (**КВС,1**), а на втором шаге вместо значений температуры на верхней границе задается условие конвективно-го теплообмена (рис. 2, б), обеспечивающее здесь охлаждение тела. Параметры этого краевого условия (коэффициент теплообмена и температура окружающей среды) постоянны на втором временном шаге, что определяется командой **КВС,0**.

После решения задачи в постпроцессоре POST26 программа строит графики зависимости температуры от времени в характерных точках $\mathbf{x}_1$ и $\mathbf{x}_2$. Эти зависимости приведены на рис. 3, а), где квадратиками отмечен график для $\theta_h(\mathbf{x}_1, t)$, а крестиками - для $\theta_h(\mathbf{x}_2, t)$. После выполнения двух закоментированных команд из файла программы можно получить графики зависимости от времени перемещений $u_{yh}(\mathbf{x}_1, t)$ и $u_{yh}(\mathbf{x}_2, t)$. Эти кривые приведены на рис. 3, б), где аналогично квадратиками отмечен график для $u_{yh}(\mathbf{x}_1, t)$, а крестиками - для $u_{yh}(\mathbf{x}_2, t)$.

(а)



(б)




Рис. 3. Температура и перемещения в характеристических точках: (а) – $\theta_h(\mathbf{x}_j, t)$; (б) – $u_{yh}(\mathbf{x}_j, t)$

Как видно из рис. 3, на этапе охлаждения температура на верхней границе спадает значительно быстрее, чем уменьшаются перемещения u_y . Кроме того, внутри тела температура и смещения постепенно выравниваются, причем температура и перемещения внутри тела могут на этапе охлаждения даже возрасти на некотором промежутке времени.

Далее в интерактивном режиме в основном постпроцессоре **POST1** можно посмотреть картинки распределения различных узловых величин NSOL для каждого временного подшага. Для этого нужно вначале из файла результатов считать

данные для выбранного подшага, например, следующим образом: *General PostProc -> Read Results -> By Pick ->* выбрать значение подшага по времени *-> Read -> Close*. Затем можно уже выводить распределения интересующих величин NSOL обычным образом. Например, *General PostProc -> Plot Results -> Contour plot-> Nodal Solution -> DOF Solution -> Nodal Temperature* – для поля температуры; *General Postproc -> Plot Results -> Vector Plo -> Predefined -> Flux&gradient -> Thermal flux TF* – для векторного показа потока тепла; и т.д.

Так, на рис. 4 приведены некоторые результаты для времени окончания первого временного шага $t = TES1$, т.е. в конце этапа нагрева: а) - распределение поля температуры, б) - векторный показ поля потока тепла. Аналогичные результаты на завершающем этапе мониторинга процесса охлаждения $t = TES2$ показаны на рис. 5.




Рис. 4. Результаты в конце первого шага: (а) – температура; (б) – вектор потока тепла




Рис. 5. Результаты в конце второго шага: (а) – температура; (б) – вектор потока тепла

Эти и другие рисунки наглядно демонстрируют распространение тепла в теле. Нужно только учитывать, что цветовая палитра на разных картинках соответствует каждый раз совершенно различным диапазонам значений.

Если в программе **FE_Mod_CP_2.inp** снять знак комментария в строке

```
! TOFFST, 0
```

т.е. выполнить команду **TOFFST, 0**, то будут проведены расчеты задачи о температурных напряжениях. Если же дополнительно снять комментарии в строках

```
! KAPPA=1e-6
```

```
! MP, DENS, 1, RHO1*KAPPA
```

```
! MP, C, 1, C1/KAPPA
```

то будет решаться задача термоупругости с частичной связанностью, когда поле температуры не зависит от механических полей, а для перемещений используется уравнение квазистатики. Расчеты показывают, что для температуры и перемещений в контрольных точках в обоих вариантах будут получены практически такие же результаты, т.е. результирующие кривые для $\theta_h(\mathbf{x}_j, t)$ и $u_{yh}(\mathbf{x}_j, t)$ визуально будут совпадать.

Наконец, отметим, что точность получаемых результатов здесь зависит от многих параметров. Максимальный размер элементов **DMESH** определяет погрешности дискретизации по пространственным переменным, и при использовании квадратичных по каноническим координатам конечных элементов **PLANE223** теоретическая погрешность порядка $O(DMESH^2)$. Подшаги по времени Δt_j дают оценки погрешности дискретизации по времени, аналогичные

оценкам погрешности дискретизации по пространственным переменным. Эти оценки справедливы лишь при наличии у решения ограниченных производных требуемой степени гладкости. Однако для областей с угловыми точками и со сменой типов граничных условий решения могут иметь локальные сингулярности, т.е. у них могут не существовать даже первые производные, ограниченные в области.

Кроме того на точность вычислений влияют и параметры численной схемы модифицированного метода Ньюмарка интегрирования по времени. В программе используются параметры по умолчанию, соответствующие команде **TINTP,0.005**. Но при больших осцилляциях решения можно увеличить параметр численного затухания схемы или изменить другие значения параметров в команде **TINTP**.

Результаты проведенных вычислительных экспериментов с варьированием размеров конечных элементов, подшагов по времени и параметров модифицированной схемы Ньюмарка показывают достаточную для модельного примера точность полей температуры и перемещений при выбранных значениях.

Индивидуальные задания – тела в форме букв.

Требуется рассчитать температурные напряжения, вызываемые температурным воздействием, для двумерной конструкции в форме буквы, указанной ниже. Рассмотреть нестационарную связанную задачу термоупругости, а также ее упрощенные модели аналогично модельному примеру. В качестве входных данных взять значения, аналогичные данным из модельного примера. Геометрические размеры областей следует выбрать самостоятельно.

Используйте командный режим программы ANSYS для создания твердотельной и конечно-элементной моделей, задания граничных условий, решения задачи и просмотра заранее определенных результатов в построителе **POST26**. Интерактивный режим ANSYS используйте для просмотра результатов в построителе **POST1** и визуализации твердотельной и конечно-элементной моделей.

Дайте анализ сходимости при различной плотности конечно-элементной сетки, значений подшагов по времени и параметров модифицированной схемы Ньюмарка. Приведите рабочий вариант конечно-элементной сетки с граничными условиями для разных шагов по времени. Проанализируйте результаты (графики зависимостей температуры и перемещений от времени в характерных точках, распределения полей температуры в деформированном состоянии, векторный показ поля потока тепла) и оформите отчет.

Отчет должен содержать данные об исполнителе, описание задачи, а также указанные выше результаты, полученные с помощью вычислительного комплекса ANSYS.

Варианты заданий лабораторной работы № 2

Ниже приводятся номера заданий и виды области (в формах заглавных букв латинского алфавита):

1 – A, 2- B, 3 – D, 4 – E, 5 – F, 6 – K, 7 – M, 8 – N, 9 – S, 10 – T, 11 – X, 12 – Y, 13 – Z, 14 – W, 15 – H.